Lecture 3b: Time-Domain Analysis of Continuous-Time Systems with Maxima

Dr.-Ing. Sudchai Boonto
Assistant Professor

Department of Control System and Instrumentation Engineering
King Mongkut's Unniversity of Technology Thonburi
Thailand

Outline

- Maxima Codes: The goodness of Maxims is the symbolic based computation with pretty given solutions. Moreover, it is an opensource software.

Zero-Input Response $y_0(t)$

To solve a differential equation, you can use a command ode2 to solve the equation. In Maxima, we use Ctrl+Enter to get a solution. For an continuous-time LTI system specified by the differential equation $(D^2 + 4D + k)y(t) = (3D + 5)f(t)$ determine the zero-input component of the response if the initial conditions are $y_0(0) = 3$, and $\dot{y}_0(0) = -7$ for two values of k: (a) 3 (b) 4 (c) 40.

Solution: part (a) $k = 3$

```
eq1: 'diff(y,t,2) + 4*'diff(y,t,1) + 3*y = 0;
```

$$\frac{d^2}{dt^2}y + 4\left(\frac{d}{dt}y\right) + 3y = 0$$

```
sol1: ode2(eq1,y,t);
```

$$y = \%k1\,\%e^{-t} + \%k2\,\%e^{-3t}$$

```
ps1: ic2(sol1,t=0,y=3,'diff(y,t)=-7);
```

$$y = \%e^{-t} + 2\%e^{-3t}$$

Zero-Input Response $y_0(t)$

part (b) $k = 4$

```
eq2: 'diff(y,t,2) + 4*'diff(y,t,1) + 4*y = 0;
```

$$\frac{d^2}{dt^2} y + 4 \left(\frac{d}{dt} y \right) + 4y = 0$$

```
sol2: ode2(eq2,y,t);
```

$$y = (\%k2t + \%k1)\ \%e^{-2t}$$

```
ps2: ic2(sol2,t=0,y=3,'diff(y,t)=-7);
```

$$y = (3 - t)\ \%e^{-2t}$$

Zero-Input Response $y_0(t)$

part (c) $k = 40$

```
eq3: 'diff(y,t,2) + 4*'diff(y,t,1) + 40*y = 0;
```

$$\frac{d^2}{dt^2}y + 4\left(\frac{d}{dt}y\right) + 40y = 0$$

```
sol3: ode2(eq3,y,t);
```

$$y = \%e^{-2t}\left(\%k1\,\sin\left(6t\right) + \%k2\,\cos\left(6t\right)\right)$$

```
ps3: ic2(sol3,t=0,y=3,'diff(y,t)=-7);
```

$$y = \%e^{-2t}\left(3\cos\left(6t\right) - \frac{\sin\left(6t\right)}{6}\right)$$

```
ratsimp(\%);
```

$$y = -\frac{\%e^{-2t}\left(\sin\left(6t\right) - 18\cos\left(6t\right)\right)}{6}$$

Zero-Input Response $y_0(t)$

To plot y respect to t, we can use following commands:

```
/* define a variable to obtain a left hand side value */
/* use ps3 as an example */
us: rhs(ps3)
plot2d(us,[t,0,10]);
```



Zero-State Response $y_s(t)$

For an LTI system specified by the differential equation

$$(D^2 + 3D + 2)y(t) = Df(t)$$

To calculate the zero-state response, we can use Maxima command set listed below:

$$h(t) = b_n\delta + P(D)y_n(t)u(t)$$

In this case $b_n = 0$ and the initial values of $y_n(t)$ are $y_n(0^-) = 0$ and $\dot{y}(0^-) = 1$.

```
eq4: 'diff(y,t,2) + 3*'diff(y,t,1)+2*y=0;
gs4: ode2(eq4,y,t);
ps4: ic2(gs4,t=0,y=0,'diff(y,t)=1);
Dy: diff(ps4,t);
```

To obtain $\frac{d}{dt}y = 2\%e^{-2t} - \%e^{-t}$. Therefore

$$h(t) = 0 + (2e^{-2t} - e^{-t})u(t)$$

Zero-State Response $y_s(t)$

If $f(t) = 10e^{-3t}$ we have

```
eq1: 'diff(y,t,2) + 3*'diff(y,t,1)+2*y=0;
gs1: ode2(eq1,y,t);
ps1: ic2(gs1,t=0,y=0,'diff(y,t)=1);
y1: diff(ps1,t);
ytau: subst(tau,t,rhs(y1));
ft : 10*exp(-3*(t-tau))

/* convolution of f(t) and y_s(t) */
assume(t>0);
result1: integrate(ytau*ft,tau,0,t);
/* Use command 'ratsimp' to simplify the equation */
```

The last line will result

$$y_s(t) = -5e^{-t} + 20e^{-2t} - 15e^{-3t},\ t \geq 0$$

Finally, the total response is $y(t) = y_0(t) + y_s(t)$.

Classical Method

Solve the differential equation

$$(D^2 + 3D + 2)y(t) = Df(t)$$

for the input $f(t) = 5t + 3$ if $y(0^+) = 2$ and $\dot{y}(0^+) = 3$.

```
Df: diff(5*t+3,t);
eq5: 'diff(y,t,2) + 3*'diff(y,t,1)+2*y =Df;
gs5: ode2(eq5,y,t);
ps5: ic2(gs5,t=0,y=2,'diff(y,t)=3);
```

The last line will show

$$y(t) = 2\%e^{-t} - \frac{5\%e^{-2t}}{2} + \frac{5}{2}$$

Reference

1. Edwin L. Woollet, *Maxima by Example*, `http://web.csulb.edu/~woollett/`.
2. Maxima Development Team: Maxima Reference Manual V.5.39. 2016.